
Nach Abschluss einer Entwicklung muss die Datenbank dokumentiert werden, um einerseits die für spätere Anpassungen notwendigen Informationen zur Verfügung zu haben oder um eine Basis für die Bearbeitung durch andere Entwickler zu schaffen. Insbesondere für den Modulbereich benötigen diese Arbeiten extrem viel Zeit, da Sie jedes Modul separat öffnen, Deklarationen und Prozeduren einzeln anzeigen, den Code herauskopieren und die Ergebnisse beispielsweise in ein Word-Dokument übertragen müssen. Aber auch während der Entwicklung müssen Sie immer wieder nach Einzelheiten in den Modulen suchen. Dazu ist zunächst in die VBA-Entwicklungsumgebung zu wechseln, ein Modul zu öffnen und die Suchfunktion zu bemühen. In diesem zweiteiligen Tipp der Woche stellen wir Ihnen eine Lösung vor, die diese Aufgabe automatisiert. Als Ergebnis erhalten Sie eine Tabelle "ModInfos", die Ihnen alle Module und die darin enthaltenen Deklarationen und Prozeduren als Basis für eine Suche oder einen Bericht zur Verfügung stellt.

Legen Sie zunächst in der zu dokumentierenden Datenbank eine neue Tabelle "ModInfos" mit den folgenden Feldern an:

ModName, Dateityp "Text"
SubName, Dateityp "Text"
SubText, Dateityp "Memo"

Ein Primärschlüssel ist für diese Tabelle nicht notwendig.

Anschließend erstellen Sie ein neues Modul "modModInfos" und kopieren die folgenden Deklarationen, Funktionen und Prozeduren in dieses Modul:

Public arrModules() As String
Public intModulesCnt As Integer

Public arrSubs() As String
Public intSubsCnt As Integer

Function ModulenamesToArray() _
         As Long
  Dim db As DAO.Database
  Dim objDocs As Object
  Dim objDoc As Object

  On Error Resume Next
  Erase arrModules()
  intModulesCnt = 0
  Set db = CurrentDb()
  Set objDocs = db.Containers("Modules").Documents
  For Each objDoc In objDocs
    If Left$(objDoc.Name, 1) <> "~" And _
       Left$(objDoc.Name, 4) <> "MSys" Then
      intModulesCnt = intModulesCnt + 1
      ReDim Preserve arrModules(intModulesCnt)
      arrModules(intModulesCnt) = objDoc.Name
    End If
  Next
  Set objDocs = Nothing
  Set objDoc = Nothing
  ModulenamesToArray = intModulesCnt

End Function

Function SubnamesToArray(strModName As String) _
         As Long
  Dim objModule As Module
  Dim lngLine As Long, lngNumLines As Long
  Dim lngProcType As Long, lngStart As Long, _
      lngEnd As Long
  Dim strProcName As String, strLast As String

  On Error GoTo Func_Error
  SubnamesToArray = 0
  Application.Echo False
  DoCmd.OpenModule strModName
  DoCmd.Minimize
  Application.Echo True
  Set objModule = Modules(strModName)
  Erase arrSubs()
  intSubsCnt = 1
  With objModule
    ReDim Preserve arrSubs(intSubsCnt)
    'Anzahl Zeilen im Deklarationsbereich
    lngLine = .CountOfDeclarationLines
    'Anzahl Zeilen des Moduls
    lngNumLines = .CountOfLines
    If lngNumLines = 0 Then
      arrSubs(intSubsCnt) = "<Leer>"
    Else
      arrSubs(intSubsCnt) = "Deklarationen"
    End If
    lngLine = lngLine + 1 'Erste Zeile des Moduls mit Subs
    strLast = ""
    'Name der Prozedur
    strProcName = .ProcOfLine(lngLine, lngProcType)
    While strProcName <> strLast
      strLast = strProcName
      intSubsCnt = intSubsCnt + 1
      ReDim Preserve arrSubs(intSubsCnt)
      arrSubs(intSubsCnt) = strProcName
      lngStart = .ProcBodyLine(strProcName, lngProcType)
      lngEnd = lngStart + _
               .ProcCountLines(strProcName, lngProcType)
      lngLine = lngEnd + 1 'Nächste Sub
      'Name der Prozedur
      strProcName = .ProcOfLine(lngLine, lngProcType)
    Wend
  End With
  DoCmd.Close acModule, strModName
  Set objModule = Nothing
  SubnamesToArray = intSubsCnt

Func_End:
  Exit Function

Func_Error:
  Application.Echo True
  Beep
  MsgBox "SubnamesToArray: " & Err.Description
  DoEvents
  Resume Func_End

End Function

Function GetSub(strModName As String, _
                strProcName As String)
  Dim objModule As Module
  Dim lngStart As Long, lngBody As Long
  Dim lngNumLines As Long, lngProcType As Long
  Dim strCode As String

  On Error Resume Next
  GetSub = "./."
  If strProcName = "" Or _
     strProcName = "<Leer>" Then Exit Function

  Application.Echo False
  DoCmd.OpenModule strModName
  DoCmd.Minimize
  Application.Echo True
  If Err <> 0 Then
    Beep
    MsgBox "GetSub: " & Err.Description
    Exit Function
  End If
  Set objModule = Modules(strModName)
  With objModule
    If strProcName = "Deklarationen" Then
      lngBody = 1
      lngNumLines = .CountOfDeclarationLines
    Else
      lngProcType = 0
      Do
        Err = 0
        lngStart = .ProcStartLine(strProcName, _
                    lngProcType)
        If Err <> 0 Then lngProcType = lngProcType + 1
      Loop Until Err = 0 Or lngProcType > 3
      lngBody = .ProcBodyLine(strProcName, lngProcType)
      lngNumLines = .ProcCountLines(strProcName, _
                    lngProcType) - (lngBody - lngStart)
    End If
    strCode = "<Leer>"
    If lngNumLines = 0 Then GoTo Proc_Ende
    strCode = objModule.Lines(lngBody, lngNumLines)
  End With
  While Right$(strCode, 4) = vbCrLf & vbCrLf
    strCode = Left$(strCode, Len(strCode) - 2)
  Wend
  While Right$(strCode, 2) <> vbCrLf
    strCode = strCode & vbCrLf
  Wend

Proc_Ende:
  GetSub = strCode
  DoCmd.Close acModule, strModName
  Set objModule = Nothing
  DoEvents

End Function

Die genaue Arbeitsweise dieser Funktionen erläutern wir im zweiten Teil dieses Tipps, in dem wir auch eine Prozedur zur Überragung der Modulinfos in eine Tabelle vorstellen - hier nur kurz die Aufgaben der einzelnen Funktionen:

· "ModulnamesToArray()" liest die Namen aller vorhandenen Module und Klassenbibliotheken in ein globales Array "arrModules" ein, das Funktionsergebnis ist die Anzahl Module in der Datenbank.

· "SubnamesToArray()" liest die Namen aller normalen Prozeduren, der Funktionen und der Property Let/Get/Set-Prozeduren für ein bestimmtes, per Parameter "strModName" spezifiziertes Modul in das globale Array "arrSubs" ein, das Ergebnis ist die Anzahl aller normalen Prozeduren, der Funktionen und der Property Let/Get/Set-Prozeduren im Modul.

· Der erste Eintrag dieses Arrays ist ggf. "Deklarationen", wenn das Modul oder die Klassenbibliothek einen Deklarationsbereich enthält.

· Den Text einer normalen Prozedur, einer Funktion oder einer Property Let/Get/Set-Prozedur liest schließlich die Funktion "GetSub" aus, der als Parameter "strModName" der Name des Moduls und als Parameter "strSubName" der Name der normalen Prozedur, einer Funktion oder eine Property Let/Get/Set-Prozedur oder "Deklarationen" zum Auslesen des Deklarationsbereiches mitgeteilt wird.

Im ersten Teil dieses Tipps hatten wir die Basis-Funktionen für eine Lösung zur Dokumentation von Modulen vorgestellt - hier noch einmal kurz die einzelnen Funktionen in der Zusammenfassung: "ModulnamesToArray()" liest die Namen aller vorhandenen Module und Klassenbibliotheken in ein globales Array "arrModules" ein, das Funktionsergebnis ist die Anzahl Module in der Datenbank. "SubnamesToArray()" liest die Namen aller normalen Prozeduren, der Funktionen und der Property Let/Get/Set-Prozeduren für ein bestimmtes, per Parameter "strModName" spezifiziertes Modul in das globale Array "arrSubs" ein, das Ergebnis ist die Anzahl aller normalen Prozeduren, der Funktionen und der Property Let/Get/Set-Prozeduren im Modul. Der erste Eintrag dieses Arrays ist ggf. "Deklarationen", wenn das Modul oder die Klassenbibliothek einen Deklarationsbereich enthält. Den Text einer normalen Prozedur, einer Funktion oder eine Property Let/Get/Set-Prozedur liest schließlich die Funktion "GetSub" aus, der als Parameter "strModName" der Name des Moduls und als Parameter "strSubName" der Name der normalen Prozedur, einer Funktion oder eine Property Let/Get/Set-Prozedur oder "Deklarationen" zum Auslesen des Deklarationsbereiches mitgeteilt wird.

Basierend darauf lässt sich eine Prozedur zusammensetzen, die zunächst per "ModulnamesToArray()" die Namen aller Module in das globale Array "arrModules" einliest und die Anzahl Module als Ergebnis liefert. Darauf baut eine Schleife auf, die alle Module der Reihe nach durchgeht und die Prozedurnamen pro Modul ins Array "arrSubs" überträgt und die Anzahl Routinen als Ergebnis liefert. In einer zweiten Schleife wird dann der Deklarationsbereich sowie der zu den einzelnen Routinen gehörende Code per "GetSub()" ausgelesen. Modulname, Deklarationsbereich und Code wird dann in eine Tabelle "ModuleInfos" geschrieben, die als Basis für eine Suche oder einen Bericht dienen kann.

Kopieren Sie zunächst zur Vervollständigung der Lösung die folgende Prozedur "ModInfosToTable" in das in der letzten Ausgabe angelegte Modul "modModuleInfos":

Sub ModInfosToTable()
  Dim db As DAO.Database, rs As DAO.Recordset
  Dim intCntModules As Integer, intCntSubs As Integer
  Dim I&, J&, strText As String

  On Error Resume Next
  DoCmd.SetWarnings False
  DoCmd.RunSQL "delete * from ModInfos;", False
  DoCmd.SetWarnings True
  DoEvents
  DBEngine.Idle

  intCntModules = ModulenamesToArray()
  If intCntModules = 0 Then Exit Sub

  Set db = CurrentDb()
  Set rs = db.OpenRecordset("ModInfos", dbOpenDynaset)
  For I = 1 To intCntModules
    intCntSubs = SubnamesToArray(arrModules(I))
    If intCntSubs > 0 Then
      For J = 1 To intCntSubs
        strText = GetSub(arrModules(I), arrSubs(J))
        rs.AddNew
        rs("ModName") = arrModules(I)
        rs("SubName") = arrSubs(J)
        rs("SubText") = strText
        rs.Update
      Next J 'Nächste Sub
    End If
  Next I 'Nächstes Modul
  rs.Close

  Set rs = Nothing
  Set db = Nothing

  Beep
  DoCmd.OpenTable "ModInfos"
  MsgBox "Fertig...", vbOKOnly + vbInformation

End Sub


Diese Prozedur, die Sie einfach im Direktbereich der VBA-Entwicklungsumgebung mit der Eingabe

ModInfosToTable

aufrufen, führt die oben umrissenen Schritte durch und schreibt die Infos in die Tabelle "ModuleInfos".

Zur Verwaltung und somit auch zum Auslesen der Modulinfos stellt Access das Objekt "Module" bereit. Leider steht hier keine spezielle Auflistung für den gezielten Zugriff auf Deklarationen, Prozeduren und Funktionen zur Verfügung. Vielmehr muss hier ausschließlich "zeilenorientiert" gearbeitet werden. Dazu stellt man sich das Modul am besten als untereinander angeordnete Prozeduren und Funktionen vor, denen der Deklarationsteil vorangestellt ist. Um das zu verdeutlichen, wählen Sie in der VBA-Entwicklungsumgebung das Menü Extras-Optionen an und aktivieren Sie die Optionen Standardmäßig Ganzes Modul Anzeigen und Prozedurtrennlinie. Die nun sichtbaren Zeilen sind aus Sicht des Objektmodells beginnend mit "1" (in der Regel die Zeile mit "Option Compare Database") bis "n" (letzte Zeile hinter der letzten Prozedur/Funktion) fortlaufend durchnummeriert. Um auf eine bestimmte Prozedur oder Funktion zuzugreifen, müssen Sie sich nun durch diese Zeilen "hindurchhangeln": Die Eigenschaft "CountOfDeclarationLines" gibt zunächst Auskunft, wie viele Zeilen der Deklarationsbereich umfasst. Die darauf folgende Zeile ist der Prozedur/Funktionskopf der ersten VBA-Routine. Den Namen dieser Routine liefert die Funktion "ProcOfLine(lngLine, lngProcType)". Der Parameter "lngLine" spezifiziert die Nummer einer Zeile, zu der der Name der dazugehörenden Prozedur oder Funktion geliefert werden soll. "lngLine" muss nicht zwingend auf den Kopf der Routine verweisen, es kann auch eine Zeile angegeben werden, die irgendwo zwischen "Sub/Function/Property" und "End Sub/End Function/End Property" liegt. Der Parameter "lngProcType" liefert den Typ der VBA-Routine wie folgt:

vbext_pk_Proc/0: Normale Prozedur/Funktion
vbext_pk_Let/1: Property Let-Prozedur
vbext_pk_Set/2: Property Set-Referenz
vbext_pk_Get/3: Property Get-Funktion



Mit dem über "ProcOfLine" ermittelten Namen muss nun die genaue Start- und Endzeile der Routine ermittelt werden:

lngStart = .ProcBodyLine(strProcName, lngProcType)
lngEnd = lngStart + _
         .ProcCountLines(strProcName, lngProcType)

"ProcBodyLine" liefert die Nummer der Zeile, in der der Kopf der Routine (Sub, Function, Property) enthalten ist, "ProcCountLines" liefert die Anzahl Zeilen, die die Routine umfasst. "lngStart + lngEnd + 1" liefert dann die Zeilennummer, in der die nächste Routine beginnt, deren Namen und Lage dann wieder per "ProcOfLine", "ProcBodyLine" und "ProcCountLines" bestimmt wird.

Eventuell fragen Sie sich, warum nach "ProcOfLine" zur Ermittlung des Namens einer an Zeile "x" liegenden Routine noch einmal "ProcBodyLine" zur Ermittlung der Startzeile aufgerufen werden muss. Vor einer Routine befindet sich generell immer eine Leerzeile, aber es können sich auch mehrere Zeilen vor der Routine befinden. Wenn Sie diese Zeile oder eine dieser Zeilen bei "ProcOfLine" als Parameter angeben, erhalten Sie zwar den korrekten Namen der Routine, weil die Leerzeile zu ihr gehört, die eigentliche Routine beginnt aber mit ihrem "Sub"-, "Function"- oder "Property"-Kopf erst eine oder mehrere Zeilen darunter. Um also Start- und Endzeile einer Routine exakt bestimmen zu können, müssen immer die drei Funktionen "ProcOfLine", "ProcBodyLine" und "ProcCountLines" in Kombination eingesetzt werden.

 FRAGE:  Die in Access Weekly vom 21.7.2006 gezeigte Lösung "VBA-Module automatisch dokumentieren" ist wirklich eine nützliche Lösung für das Administrieren von Datenbanken. Was mich allerdings noch stört ist die Beschränkung auf die aktuelle Datenbank. Meine Versuche, sie auch auf externe Datenbanken (z. B. mit OpenDatabase(strExterneDatenbank) usw.) anzupassen, führten leider nicht zum gewünschten Ergebnis. Das lag m.E. am "DoCmd.OpenModule", der nicht die externen MDBs anspricht bzw. ansprechen kann. Haben Sie eine Lösung, mit der die Prozeduren und Funktionen einer externen Datenbank ausgelesen werden können? Prinzipiell könnte ich eine MDE als Verweis einbinden, um dadurch die Inhalte zu bekommen, aber vielleicht gibt es ja eine Lösung, aus einer MDB auf die VBA-Projekte zuzugreifen, so wie es auch bei Ihrem Tipp zu den Objekten war.
T. Eicker

 ANTWORT:  Leider ist der Weg über eine MDE die einzig mögliche Lösung. Der Zugriff auf sämtliche VBA-Objekte ist nur für die aktuell geöffnete, nicht aber für eine extern geöffnete Datenbank möglich.

